// File: contracts/TokenLockup.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.19;

import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

contract TokenLockup is Ownable {
    using SafeERC20 for IERC20;

    /// @dev Keeps track of addresses that have administrative privileges in the contract
    ///      Admins have the capability to update the `lockupDuration` value
    mapping(address => bool) public admins;

    struct LockInfo {
        /// @notice The amount locked up in the contract
        uint256 amount;
        /// @notice The time in seconds in which the lockup tokens can be withdrawn
        uint256 unlockTime;
    }
    /// @dev Maps user to their locks, with unlockTime as key and amount as value
    ///      Each address is associated with a `LockInfo` structure that contains details
    ///      about the amount of tokens locked and the timestamp when they can be unlocked.
    mapping(address => LockInfo[]) public userLocks;

    /// @notice The ERC20 token used for locking in this contract
    IERC20 public token;

    /// @notice The lock up duration in seconds for the token
    /// @dev Added with block.timestamp (if it's the first time the user is locking)
    uint256 public lockupDuration;

    /// @notice Emitted when a new admin is added to the contract
    /// @param admin The address of the new admin
    event AdminAdded(address indexed admin);

    /// @notice Emitted when an admin is removed from the contract
    /// @param admin The address of the admin removed
    event AdminRemoved(address indexed admin);

    /// @notice Emitted when a user lock up tokens
    /// @param user The user who has locked up the tokens
    /// @param amount The amount which the user has locked up
    /// @param unlockTime The time in seconds in which the tokens will be available to withdraw
    event TokensLocked(
        address indexed user,
        uint256 amount,
        uint256 unlockTime
    );

    /// @notice Emitted when a user withdraw the tokens (after the lockup time)
    /// @param user The user has withdrawn the tokens
    /// @param amount The amount which the user has withdrawn
    event TokensWithdrawn(address indexed user, uint256 amount);

    /// @notice Emitted when an admin changes the lockupDuration of the contract
    /// @param newDuration The new lockup duration for the contract
    event LockupDurationUpdated(uint256 newDuration);

    /// @notice Duration value must be > 1 day and < 1460 days (4 years)
    error InvalidDuration();

    /// @notice Wrong amount sent to lock / insufficient amount to withdraw
    error InsufficientAmount();

    /// @notice Only owner and admin can call the function
    error NotAuthorized();

    /// @dev Initializes the contract with the given token address and lockup duration
    ///      The lockup duration must be between 1 day and 10 years
    /// @param _tokenAddress The ERC20 token address to be locked up
    /// @param _lockupDuration The lock duration for the ERC20 token
    constructor(address _tokenAddress, uint256 _lockupDuration) {
        if (_lockupDuration < 1 days || _lockupDuration > 1460 days)
            revert InvalidDuration();

        token = IERC20(_tokenAddress);
        lockupDuration = _lockupDuration;
    }

    /// @notice Modifies function to be called by owner / admins only
    modifier onlyAdmin() {
        _validateAuth();
        _;
    }

    /// @notice Validates that the caller has appropriate authorization
    function _validateAuth() internal view {
        if (msg.sender != owner() && !admins[msg.sender])
            revert NotAuthorized();
    }

    /// @notice Locks the specified amount of tokens in the contract for the lockup duration
    /// @param amount The number of tokens to lock up
    function lock(uint256 amount) external {
        if (amount == 0) revert InsufficientAmount();
        uint256 unlockTime = block.timestamp + lockupDuration;
        userLocks[msg.sender].push(LockInfo(amount, unlockTime));

        emit TokensLocked(msg.sender, amount, unlockTime);

        // Transfer tokens from user to this contract
        token.safeTransferFrom(msg.sender, address(this), amount);
    }

    /// @notice Withdraws locked tokens if the lockup duration has passed
    function withdraw() external {
        LockInfo[] storage locks = userLocks[msg.sender];
        uint256 totalAmount = 0;
        uint256 originalLength = locks.length;

        // Temporary array to hold non-eligible locks
        LockInfo[] memory nonWithdrawableLocks = new LockInfo[](locks.length);
        uint256 nonWithdrawableCount = 0;

        for (uint256 i = 0; i < originalLength; i++) {
            if (block.timestamp >= locks[i].unlockTime) {
                // Lock is eligible for withdrawal
                totalAmount += locks[i].amount;
            } else {
                // Lock is not yet withdrawable, so we re-add it to the temporary array
                nonWithdrawableLocks[nonWithdrawableCount] = locks[i];
                nonWithdrawableCount++;
            }
        }

        if (totalAmount == 0) revert InsufficientAmount();

        // Reset and repopulate user locks with only non-withdrawable locks
        delete userLocks[msg.sender];
        for (uint256 i = 0; i < nonWithdrawableCount; i++) {
            userLocks[msg.sender].push(nonWithdrawableLocks[i]);
        }

        emit TokensWithdrawn(msg.sender, totalAmount);
        token.safeTransfer(msg.sender, totalAmount);
    }

    /// @notice Emergency withdraw function callable by the owner/admin to withdraw all tokens from a user's locks
    /// @param user Address of the user to withdraw all tokens
    function emergencyWithdraw(address user) external onlyAdmin {
        uint256 totalAmount = 0;
        LockInfo[] storage locks = userLocks[user];

        for (uint256 i = 0; i < locks.length; i++) {
            totalAmount += locks[i].amount;
        }

        if (totalAmount == 0) revert InsufficientAmount();

        delete userLocks[user];

        emit TokensWithdrawn(user, totalAmount);
        token.safeTransfer(user, totalAmount);
    }

    /// @notice Calculates and returns the total amount of tokens locked and unlockable for a given user
    /// @param user The address of the user whose lock status is being queried
    /// @return totalLocked The total amount of tokens that are still locked for the user
    /// @return totalUnlockable The total amount of tokens that the user can currently unlock
    function getAmounts(address user)
        public
        view
        returns (uint256 totalLocked, uint256 totalUnlockable)
    {
        LockInfo[] storage locks = userLocks[user];
        totalLocked = 0;
        totalUnlockable = 0;

        for (uint256 i = 0; i < locks.length; i++) {
            LockInfo memory lockInfo = locks[i];
            totalLocked += lockInfo.amount;

            if (block.timestamp >= lockInfo.unlockTime) {
                totalUnlockable += lockInfo.amount;
            }
        }

        return (totalLocked, totalUnlockable);
    }

    /// @notice Returns the next unlock time and the corresponding amount for a given user's locked tokens
    /// @dev The function does not assume that the user's lock entries are in chronological order (in the event when the lock duration changes)
    /// @param user The address of the user whose lock information is being queried
    /// @return nextUnlockTime The timestamp of the next unlock event. Returns 0 if there are no future unlocks
    /// @return amount The amount of tokens that will be unlocked at the `nextUnlockTime`
    ///                Returns 0 if there are no future unlocks
    function getNextUnlockTimeAndAmount(address user)
        public
        view
        returns (uint256 nextUnlockTime, uint256 amount)
    {
        LockInfo[] storage locks = userLocks[user];
        nextUnlockTime = type(uint256).max;

        for (uint256 i = 0; i < locks.length; i++) {
            if (
                locks[i].unlockTime > block.timestamp &&
                locks[i].unlockTime < nextUnlockTime
            ) {
                nextUnlockTime = locks[i].unlockTime;
                amount = locks[i].amount;
            }
        }

        // There are no future unlocks
        if (nextUnlockTime == type(uint256).max) {
            return (0, 0);
        }
        return (nextUnlockTime, amount);
    }

    /// @notice Sets the lockup duration
    /// @param _duration New lockup duration in seconds
    /// @dev This function can only be called by the admin
    /// @custom:role-only-owner-admin
    function setLockupDuration(uint256 _duration) external onlyAdmin {
        // Duration must be more than 1 day and less than 4 years
        if (_duration < 1 days || _duration > 1460 days)
            revert InvalidDuration();

        lockupDuration = _duration;

        emit LockupDurationUpdated(_duration);
    }

    /// @notice Adds an admin to the contract
    /// @param admin Address of the admin to be added
    /// @custom:role-only-owner
    function addAdmin(address admin) external onlyAdmin {
        admins[admin] = true;

        emit AdminAdded(admin);
    }

    /// @notice Removes an admin from the contract
    /// @param admin Address of the admin to be removed
    /// @custom:role-only-owner
    function removeAdmin(address admin) external onlyAdmin {
        delete admins[admin];

        emit AdminRemoved(admin);
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}